

LABORATÓRIO 2 - SIMULAÇÃO SPICE DO INVERSOR

Revisão: 17/março/2017
Fernando Moraes

O objetivo deste laboratório é realizar 6 simulações com o inversor:

1. Curva DC
2. Dimensionamento
3. Carga na Saída
4. Efeito da rampa de entrada
5. Efeito de regeneração de nível lógico no inversor
6. Oscilador em Anel

Baixar os arquivos de “`wget http://www.inf.pucrs.br/~moraes/microel/lab2/lab2.tar`”. Os seguintes arquivos são utilizados (tar -xvf lab2.tar; cd lab2):

- inv0_dc.sp
- inv1_w.sp
- inv2_load.sp
- inv3_rampa.sp
- inv4_serie.sp
- inv5_anel_parcial.sp
- st65.scs [arquivo de tecnologia]
- **inv.sp**

O arquivo **inv.sp** é o principal arquivo utilizado neste laboratório, pois descreve o inversor e a respectiva tecnologia. Este arquivo é empregado em todas as simulações deste laboratório. Observar que o dimensionamento N e P são dados pelas variáveis *wp* e *wn*, as quais são definidas nos arquivos que instanciam o *inv.sp*.

```
simulator lang=spectre insensitive=no
include "st65.scs"
simulator lang=spice

*
* descrição do inversor na forma de um subcircuito
*
.subckt inv in out vcc
M1 out in vcc vcc pvtgp w=wp l=0.06
M2 out in 0 0 nvtgp w=wn l=0.06
.ends inv

vcc vcc 0 dc 1.0
```

Os arquivos com sufixo *.sp* contêm a descrição spice do circuito, enquanto que o arquivo *st65.scs* define a tecnologia que iremos empregar (65 nm). O arquivo de tecnologia deve estar no mesmo diretório do arquivo contendo a descrição spice.

Lembrar de configurar o ambiente:



source /soft64/source_gaph ; module load ic mmsim

1) SIMULAÇÃO DC DE UM INVERSOR

- Simular o arquivo **inv0_dc.sp** (*spectre* e *viva*). Esta é uma simulação DC, em que se plota o tensão de entrada (*v(iv)*, eixo x) *versus* a tensão de saída de saída (*v(out)*, eixo y).
- A descrição do circuito compreende um inversor conectado à uma capacitância de 5fF:

```
X1 iv out vcc inv
C1 out 0 5fF
```

- Na interface gráfica para cada simulação *dc* plotar o **v(out)** para cada valor de *W*, como nas figuras seguintes.




Responda:

- 1.1. No relatório, indicar o dimensionamento dos transistores em cada curva, inserindo *labels* sobre as mesmas (por exemplo: $W_n=x/W_p=x$). Explique o comportamento das curvas.
- 1.2. A figura da direita é um zoom em torno da meia excursão do sinal (0.5 V) na entrada, mostrando que a curva apontada está com a saída próxima também a 0,5 V. Qual a relação W_p/W_n neste caso? O que esta relação indica?

2) SIMULAÇÃO DE UM INVERSOR PARA DIFERENTES DIMENSIONAMENTOS DE TRANSISTOR

Simular o arquivo: **inv1 w.sp.**
Nesta simulação avalia-se o atraso de um inversor em função do dimensionamento dos transistores, variando-se o W_p de 0.2 a 1.0 μm (W_n é constante e igual a 0.2 μm). Plotar $v(\text{out})$ em relação ao tempo:



Observar no arquivo spice os comandos de medida:

```
* mede os tempos de propagacao
.measure tran pHL trig v(iv) val=0.5 td=5n rise = 1
+      targ v(out) val=0.5      fall = 1

.measure tran pLH trig v(iv) val=0.5 td=5n fall = 1
+      targ v(out) val=0.5      rise = 1

.measure tran TD param = 'pHL * 1e12'
.measure tran TS param = 'pLH * 1e12'
```

Nestes comandos estamos medindo os tempos de propagação para descida (TD) e subida (TS), em picosegundos (10^{-12})

Pede-se:

- 2.1. Fazer uma tabela para os tempos de propagação de subida e descida (arredondar para duas casas decimais). **Os dados dos comandos de medida estão no arquivo inv1_w.measure.** A coluna “diferença” corresponde à medida *diff* no arquivo measure.

Dimensão Wp (microns)	T. Descida (ps)	T. Subida (ps)	TD-TS - diferença
0,2			
0,4			
0,45			
0,5			
0,6			
0,8			
1,0			

- 2.2. Notar que apesar do transistor N manter a mesma dimensão (0.2 μm), o tempo de propagação de descida (TD) aumentou com o aumento do transistor P. Pesquisar a razão e explicar.
- 2.3. Trace no excel (ou outro programa de gráficos como *gnuplot*) a evolução dos tempos de propagação de subida/descida em função do dimensionamento dos transistores. O gráfico terá duas curvas, no eixo X o dimensionamento do transistor P, e no eixo y os tempos TD e TS. **CUIDAR: USAR NO EXCEL DISPERSÃO X-Y PARA EVITAR ERRO NO EIXO X**
- 2.4. Sendo $w_n=0.2 \mu\text{m}$, para qual tamanho de W_p/W_n os tempos de subida e descida igualem-se (ou possuem a menor diferença)? Comparar com a primeira simulação e discutir. Mostrar este ponto no gráfico da questão 2.3.


3) SIMULAÇÃO DE UM INVERSOR PARA DIFERENTES CONDIÇÕES DE CARGA

- Simular o arquivo: **inv2_load.sp**. Nesta simulação:
 - avalia-se o atraso do inversor em função da carga de saída, variando-se o CI de 1fF a 50 fF.
 - observa-se a **corrente consumida** durante o chaveamento (mesmo artifício empregado na descrição spice, fonte de tensão nula com medida de corrente sobre a mesma):


```
X1 iv out vcc inv
vii out out2
C3 out2 0 c1
```
 - Notar que neste circuito estamos medindo também a potência média consumida pelo circuito:


```
.measure cons_medio AVG POWER FROM=0n TO=25n
```
 - Os comandos de medida no arquivo de simulação já fornecem o atraso em ps e o consumo de potência em microwatt:


```
.measure tran POT param = 'cons_medio * 1e6'
.measure tran TD param = 'pHL * 1e12'
.measure tran TS param = 'pLH * 1e12'
```
 - A capacitância de entrada de um inversor com $W_n=0.2 \mu\text{m}$ e $W_p=0.45 \mu\text{m}$ é dada por:
 - Cox (capacitância unitária de óxido de silício) = E_{ox} / t_{ox} (permissividade elétrica do óxido de silício dividido pela espessura da junção do gate)
 - Nos modelos do *design kit* $t_{ox} = 1,88 \cdot 10^{-9} \text{m}$.
 - Logo, $c_{ox} = (34,53 \cdot 10^{-12} \text{ F/m}) / (1,88 \cdot 10^{-9} \text{m}) = 0,018367 \text{ F/m}^2 = 18,367 \text{ fF/um}^2$ (para a tecnologia do exemplo)
 - Assim, $C_{in} = (W_p \cdot L_p + W_n \cdot L_n) \cdot C_{ox} \rightarrow L=0.06 \mu\text{m}$, $W_n=0.2 \mu\text{m}$ e $W_p=0.45 \mu\text{m}$
 - $C_{inv} = 0,716313 \text{ fF}$**



Pede-se:


- 3.1. Plotar a tensão de entrada ($v(iv)$), as tensões de saída ($V(out)$), e as correntes em $i(vii)$, como acima. No *plot* indicar para cada curva a carga de saída utilizando *labels*.
- 3.2. Fazer uma tabela para os diferentes tempos de subida e descida, potência média consumida e o fanout equivalente. Os dados de medida são obtidos no arquivo `inv2_load.measure`.

Carga de Saída (fF)	TD (em ps)	TS (em ps)	Power - POT (em μW)	Fanout equivalente (carga de saída / C_{inv})
1				
3				
8				
15				
25				
50				

- 3.3. Trace um gráfico com a evolução dos tempos de subida e descida em função da carga de saída. A evolução do atraso é linear com a variação da carga?
- 3.4. Trace um gráfico com a evolução do consumo de potência em função da carga de saída. A evolução do consumo de potência é linear com a variação da carga?
- 3.5. Porque a energia consumida por uma porta lógica CMOS aumenta com o aumento da carga de saída? Utilizar como referência a curva de corrente obtida na simulação.
- 3.6. Qual a corrente do inversor quando o mesmo não está mudando de estado? Passe o mouse sobre a curva de corrente para obter o valor de corrente quando não há transição de estado.

(4) SIMULAÇÃO DE UM INVERSOR COM DIFERENTES RAMPAS DE ENTRADA

- Simular o arquivo **inv3_rampa.sp**. Nesta simulação, ao contrário das anteriores, gera-se apenas um arquivo de saída. Não há o comando *alter* na descrição. As curvas abaixo ilustram o efeito da rampa de entrada na saída do inversor.



Pede-se:

- 4.1. Plotar as tensões de entrada (vin1 a vin5) e as tensões de saída (o1 a o5), como acima. No *plot* indicar para cada curva qual a inclinação das rampas de entrada utilizando *labels*.
- 4.2. Fazer uma tabela para os diferentes tempos de subida e descida, conforme os exercícios anteriores. Usar como parâmetro a inclinação da entrada. Por exemplo, em `vin1 i1 0 pulse (1.2 0 1.298n 0.002N 0.002N 1N 2N)` a inclinação de subida/descida é 0.02ns/0.02ns.

Rampa (ns)	TD (em ps)	TS (em ps)
0,002		
0,02		
0,08		
0,15		
0,3		

- 4.3. Fazer um resumo explicando a influência no atraso no inversor dos seguintes parâmetros:
 - a. influência do tamanho do transistor
 - b. da carga
 - c. da rampa no atraso do inversor

Pesquise (e apresente) as equações de atraso do inversor e procure comparar as equações com o obtido na simulação.

(5) SIMULAÇÃO DE INVERSORES EM SÉRIE


- Simular o arquivo **inv4_serie.sp**.
- O arquivo spice contém 5 inversores em série:

```
X1 i a vcc inv
X2 a b vcc inv
X3 b c vcc inv
X4 c d vcc inv
X5 d e vcc inv
c1 d 0 15fF
```

- O plot abaixo ilustra a entrada do circuito ($V(i)$) e a saída dos inversores X1 ($v(a)$), X2 ($v(b)$) e X3 ($v(c)$). A entrada $v(i)$ excursiona apenas entre 0,49 e 0,51 V:

```
vin i 0 pulse (na nb 0 0.02N 0.02N 1N 2N)
.param na=0.49 nb=0.51
```

A saída do primeiro inversor (X1), no sinal 'a', provê uma excursão entre (0,39/0,054). Em X2 temos (0,21/0,92). Na saída de X3 ($V(c)$) já temos os níveis lógicos completamente restaurados. Esta simulação mostra a capacidade de regeneração dos níveis lógicos pelo inversor.



Pede-se:

- Defina a excursão mínima do sinal de entrada para que ao final de 2 inversores ($v(b)$) o sinal esteja restaurado (**tem de ser de forma empírica, aumentar gradualmente a excursão do sinal de entrada em vin com passo de 0.005 volts – 5 mV**). Apresentar uma curva para cada teste, inserindo em cada curva os valores de excursão em $v(i)$, $v(a)$, $v(vb)$.

Critério para regeneração do sinal:


0: menor que 0,05V (50 mV)

1: maior que 0,95V (950 mV)

- Por deve-se aumentar a excursão do sinal para ocorrer a regeneração dos níveis lógicos?

(6) SIMULAÇÃO DE UM ANEL DE INVERSORES

COMPLETAR o arquivo `inv5_anel_parcial.sp`, de forma que o mesmo venha a ter 15 inversores em série. O objetivo desta simulação é determinar o período de oscilação dos inversores em anel. O resultado obtido, graficamente, para o sinal de saída do último inversor está desenhado abaixo. Pelos *labels* obtidos temos o período igual a **0,242019 ns** ($2,94874 - 2,706721$) (usei *marker* → *create marker* → *horizontal marker* em 0.5 volts), correspondendo a uma frequência de 4,13 GHz.



Dois métodos de medida são utilizados. O primeiro utiliza os tempos de subida e descida dos inversores:

```
.measure tran tf      trig v(n)  val=0.5  td=1n rise = 10
+                    targ v(o)  val=0.5      fall = 10
.measure tran tr      trig v(n)  val=0.5  td=1n fall = 10
+                    targ v(o)  val=0.5      rise  = 10

.measure tran periodo param = '(tf+tr) * 1e9 * 15'
.measure tran freq    param = '1/periodo'
```

O segundo método mede a onda de saída apenas, entre 2 subidas em `v(o)`:

```
.measure tran periodo_o trig v(o)  val=0.5  td=1n rise = 2
+                    targ v(o)  val=0.5      rise = 3
```

No arquivo de medida (`inv5_anel.measure`) obtém-se:

```
Measurement Name : transient1
Analysis Type    : tran
freq             = 4.1318          → frequência de oscilação igual a 4,13 GHz
periodo          = 0.242025
periodo_o        = 2.4202e-10
tf               = 8.18305e-12
tr               = 7.95197e-12
```

Estas medidas correspondem a um tempo de propagação na descida igual a 8,18 ps e na subida 7,95 ps. Assim, o período de oscilação corresponde a 242 ps, o que confere com os dois comandos *measure* (período e período_o), mostrando que os dois métodos de medida são equivalentes.

Pede-se:

- 6.1. Utilizando o comando `.alter`, modificar o `wp` do tamanho original até 50 μm (0.45, 1, 2, 4, 6, 8, 10, 12, 14, 18, 20, 25, 30, 35, 40, 45, 50) - o `Wn` é função do `Wp`. Plotar um gráfico tendo como eixo x o dimensionamento de `wp` e no eixo y a frequência obtida. É possível chegar com 15 estágios em **5 GHz**? Argumente a sua resposta.

```
.param wp=0.45 wn='wp/2.25'
.alter
.param wp=1 wn='wp/2.25'
.alter
.....
```

FIM DO LABORATÓRIO 2